Use <CTRL>-1 (<CMD>-1 on the Mac) to insert a text cell where the cursor is.
Otherwise, just start typing where the cursor is to create an input cell.

Use <Shift>-Enter to evaluate a cell.
    End a cell with $ if you don't want to see the output of a cell.
    End with semicolon to display the cell evaluation. For a single line cell, the semicolon is added automatically
    For multiple lines in a cell, use Enter at the end of each line and <Shift>-Enter to evaluate all of them.

There are wxMaxima buttons that make it convenient to do certain things.
Use Maxima->Panes->Insert Cell and Maxima->Panes->General Math to display the buttons.

The expression % refers to the output of the last evaluation. You can refer to a previous output by, for example, %o54.

You can suppress output (end line with $) or display it (end with ; or nothing at all for a single line)

```
(%i1) 500*12$
```

```
(%i2) 500*12;
      2^10;
```
(%o2) 6000
(%o3) 1024

Evaluate some expressions. If you use integers or predefined values such as %e and %pi, wxMaxima will treat the expression as exact.

```
(%i4) 2/6;
```
(%o4) $\frac{1}{3}$

```
(%i5) 3^5;
```
(%o5) 243

```
(%i6) log(%e);
      cos(%pi);
```
(%o6) 1
(%o7) -1

```
(%i8) sqrt(120);
```
(%o8) $2\sqrt{30}$

```
(%i9) 1/4 + 1/3 + 1/5;
```
(%o9) $\frac{47}{60}$

```
(%i10) factor(60);
```
(%o10) $2^{2}\ 3\ 5$

```
(%i11) factor(100!);
```
(%o11) $2^{97}\ 3^{48}\ 5^{24}\ 7^{16}\ 11^{9}\ 13^{7}\ 17^{5}\ 19^{5}\ 23^{4}\ 29^{3}\ 31^{3}\ 37^{2}\ 41^{2}\ 43^{2}\ 47^{2}\ 53\ 59\ 61\ 67\ 71\ 73\ 79\ 83\ 89\ 97$

Using a number like 120.0 will cause a "floating point" evaluation by default. The standard precision is about 15-16 digits (standard double precision). The floating point value is normally not exact. It can represent numbers with magnitudes between approximately 10^(-324) to 10^(308).

A mixture of exact expressions and floating point is converted to floating point.

```
(%i12) sqrt(120.0);
       sqrt(120);
(%o12) 10.95445115010332
(%o13) 2 √30
```

```
(%i14) 1/4.0 + 1/3 + 1/5;
       1/4 + 1/3 + 1/5;
(%o14) .7833333333333332
(%o15) 47/60
```

You can also force evaluation to floating point with the "numer" (short for numeric) suffix.
The float function can also be used to force a floating point evaluation.

```
(%i16) 1/4 + 1/3 + 1/5, numer;
(%o16) .7833333333333332
```

```
(%i17) float(1/4 + 1/3 + 1/5);
(%o17) .7833333333333333
```

```
(%i18) sqrt(120), numer;
       %pi, numer;
       float(1/5);
(%o18) 10.95445115010332
(%o19) 3.141592653589793
(%o20) 0.2
```

You can create some *big* numbers

```
(%i21) 5^(3^4);
(%o21) 413590306276513837435704346034981426782906055450439453125
```

You can use the built-in constants. There are other constants like inf (infinity) and
minf (-infinity), which are used in integrations and limits.

Make a list by enclosing a list of comma-delimited values in square brackets,
e.g. [1, 2, %pi, float(%e^5)].

You can use float to get a floating point value of an expression.

```
(%i22) list1 : [%e, %pi, %i, %phi, %gamma];
(%o22) [%e,π,%i,φ,γ]
```

Use the append function to append another list to a list

```
(%i23) list2 : append(list1, [minf, inf]);
(%o23) [%e,π,%i,φ,γ,-∞,∞]
```

```
(%i24) float(%);
(%o24) [2.718281828459045,3.141592653589793,%i,1.618033988749895,.5772156649015329,-∞,∞]
```

```
(%i25) float(100!);
(%o25) 9.33262154439442 10^157
```

You can also create and evaluate algebraic and other expressions

```
(%i26) 1/(x-1) + 1/(x+1);
```
(%o26) $\frac{1}{x+1}+\frac{1}{x-1}$

```
(%i27) x^4 - 2*x^2 + 1;
       %, x=10;
```
(%o27) $x^4-2\,x^2+1$

(%o28) 9801

```
(%i29) sin(x)*cos(x)^2 + exp(x);
       %, x=%pi/2;
```
(%o29) $\cos(x)^2\,\sin(x)+\%e^x$

(%o30) $\%e^{\pi/2}$

Trig, inverse trig, exponential and log functions

```
(%i31) sin(0.8);
       asin(%);
(%o31) .7173560908995228
(%o32) 0.8
```

```
(%i33) theta: 0.6;
       tan(theta);
       sin(theta)/cos(theta);
       atan(%);
(%o33) 0.6
(%o34) .6841368083416923
(%o35) .6841368083416923
(%o36) 0.6
```

```
(%i37) exp(2.5);
       log(%);
(%o37) 12.18249396070348
(%o38) 2.5
```

You can do substitution using a suffix

```
(%i39) x^2 - 5*x + 7, x=a+b;
```
(%o39) $(b+a)^2-5\,(b+a)+7$

```
(%i40) %, expand;
```
(%o40) $b^2+2\,a\,b-5\,b+a^2-5\,a+7$

You can assign an expressions to a variable using : (equal sign is *not* used for assignment in wxMaxima)
We define the variable g below.

```
(%i41) g : x^2 - 5*x + 7;
```
(%o41) $x^2-5\,x+7$

```
(%i42) 2*g;
```
(%o42) $2\,(x^2-5\,x+7)$

```
(%i43) g, x=a+b;
```
(%o43) $(b+a)^2-5\,(b+a)+7$

Create a function with arguments using :=
A function with arguments allows anything to be used as one of the arguments.
An expression is fixed to the actual expression entered with the variables as written.

```
(%i44) ff(x) := sin(x)^2;
```

(%o44) $ff(x):=\sin(x)^2$

Calculate some values with this function. wxMaxima treats exact and floating point values differently.

```
(%i45) ff(5);
       ff(5.0);
       ff(%pi);
       ff(a), a=5;
```

(%o45) $\sin(5)^2$
(%o46) .9195357645382262
(%o47) 0
(%o48) $\sin(5)^2$

You can also define a function with define(). There are situations when you need to do it this way.

```
(%i49) define(ff2(x), sin(x)^2);
```

(%o49) $ff2(x):=\sin(x)^2$

```
(%i50) ff2(5.0);
```

(%o50) .9195357645382262

You can factor integers and expressions

```
(%i51) factor(100!);
```

(%o51) $2^{97}\ 3^{48}\ 5^{24}\ 7^{16}\ 11^{9}\ 13^{7}\ 17^{5}\ 19^{5}\ 23^{4}\ 29^{3}\ 31^{3}\ 37^{2}\ 41^{2}\ 43^{2}\ 47^{2}\ 53\ 59\ 61\ 67\ 71\ 73\ 79\ 83\ 89\ 97$

```
(%i52) factor(x^2-2*x-15);
```

(%o52) $(x-5)(x+3)$

Factoring can be applied to previously defined expressions and functions

```
(%i53) g : x^2 - 2*x - 15;
       h(x) := x^2 - 2*x - 15;
```

(%o53) $x^2-2x-15$
(%o54) $h(x):=x^2-2x-15$

```
(%i55) factor(g);
       factor(h(x));
```

(%o55) $(x-5)(x+3)$
(%o56) $(x-5)(x+3)$

You can add factor as a suffix modifier to an expression

```
(%i57) 100!, factor;
```

(%o57) $2^{97}\ 3^{48}\ 5^{24}\ 7^{16}\ 11^{9}\ 13^{7}\ 17^{5}\ 19^{5}\ 23^{4}\ 29^{3}\ 31^{3}\ 37^{2}\ 41^{2}\ 43^{2}\ 47^{2}\ 53\ 59\ 61\ 67\ 71\ 73\ 79\ 83\ 89\ 97$

Take the 12th derivative of exp(-x^2) and factor the resulting expression

```
(%i58) diff(exp(-x^2), x, 12), factor;
       %, x=0;
```

(%o58) $64\left(64\,x^{12}-2112\,x^{10}+23760\,x^{8}-110880\,x^{6}+207900\,x^{4}-124740\,x^{2}+10395\right)\%e^{-x^{2}}$

(%o59) 665280

The opposite of factor is expand. You can also use expand as a function or as modifier to an expression

```
(%i60) expand( (x-5)*(x+3) );
```

(%o60) $x^{2}-2\,x-15$

```
(%i61) (x-5)*(x+3), expand;
```

(%o61) $x^{2}-2\,x-15$

```
(%i62) (x-1)^10, expand;
```

(%o62) $x^{10}-10\,x^{9}+45\,x^{8}-120\,x^{7}+210\,x^{6}-252\,x^{5}+210\,x^{4}-120\,x^{3}+45\,x^{2}-10\,x+1$

```
(%i63) %, factor;
```

(%o63) $\left(x-1\right)^{10}$

```
(%i64) (x^2-9) * (x^4-1) * (x+5) * (x^2-6), expand;
```

(%o64) $x^{9}+5\,x^{8}-15\,x^{7}-75\,x^{6}+53\,x^{5}+265\,x^{4}+15\,x^{3}+75\,x^{2}-54\,x-270$

```
(%i65) %, factor;
```

(%o65) $\left(x-3\right)\left(x-1\right)\left(x+1\right)\left(x+3\right)\left(x+5\right)\left(x^{2}-6\right)\left(x^{2}+1\right)$

```
(%i66) 1/(x^2-1);
```

(%o66) $\frac{1}{x^{2}-1}$

```
(%i67) factor(%);
```

(%o67) $\frac{1}{\left(x-1\right)\left(x+1\right)}$

```
(%i68) expand(%);
```

(%o68) $\frac{1}{x^{2}-1}$

trigexpand transforms trig expressions to terms with sin(x)^n and cos(x)^n
trigreduce transforms trig expressions to terms involving sin(n*x) and cos(n*x)

```
(%i69) trigexpand(sin(4*x) + cos(4*x));
       %, trigreduce;
       %, factor;
```

(%o69) $\sin\left(x\right)^{4}-4\cos\left(x\right)\sin\left(x\right)^{3}-6\cos\left(x\right)^{2}\sin\left(x\right)^{2}+4\cos\left(x\right)^{3}\sin\left(x\right)+\cos\left(x\right)^{4}$

(%o70) $2\left(\frac{\sin\left(4\,x\right)}{4}+\frac{\sin\left(2\,x\right)}{2}\right)-2\left(\frac{\sin\left(2\,x\right)}{2}-\frac{\sin\left(4\,x\right)}{4}\right)+\frac{3\cos\left(4\,x\right)-3}{4}+\frac{\cos\left(4\,x\right)+4\cos\left(2\,x\right)+3}{8}+\frac{\cos\left(4\,x\right)-4\cos\left(2\,x\right)+3}{8}$

(%o71) $\sin\left(4\,x\right)+\cos\left(4\,x\right)$

ratsimp makes a common denominator. expand is the approximate opposite of ratsimp

```
(%i72) ratsimp(1/a + 1/b);
       ratsimp(1/a + 1/b + 1/c);
       expand(%);
```

(%o72) $\frac{b+a}{a\,b}$

(%o73) $\frac{(b+a)\,c+a\,b}{a\,b\,c}$

(%o74) $\frac{1}{c}+\frac{1}{b}+\frac{1}{a}$

```
(%i75) 1/(x-1) + 1/(x+1) + 1;
```

(%o75) $\frac{1}{x+1}+\frac{1}{x-1}+1$

```
(%i76) ratsimp(%);
       factor(%);
```

(%o76) $\frac{x^2+2\,x-1}{x^2-1}$

(%o77) $\frac{x^2+2\,x-1}{(x-1)\,(x+1)}$

************ Obscure wxMaxima topic *************

Sometimes not all the digits are displayed for huge exact numbers.

```
(%i78) 70!;
(%o78) 1197857166996989179607278372l6[41 digits]58278984531968000000000000000000
```

To show all the digits, use set_display(ascii). To go back to hiding digits, use set_display(xml). Yes, this drives me crazy too.

```
(%i79) set_display(ascii);
       70!;
(%o79)                               ascii
(%o80) 11978571669969891796072783721689098736458938142546425857555362864628009\
58278984531968000000000000000000
```

```
(%i81) set_display(xml);
       70!;
(%o81) xml
(%o82) 119785716699698917960727837216[41 digits]58278984531968000000000000000000
```